
RUNDEN IN EXCEL 
=RUNDEN(Zahl;Anzahl_Stellen) 
Mit Hilfe der Funktion =RUNDEN(Zahl;Anzahl_Stellen) können Sie Zahlen auf Vor-, Nachkommastellen 
oder auf ganze Zahlen runden. Mit Hilfe eines kleinen Tricks ist es auch möglich auf halbe Werte oder gar 
5-Räppler zu runden. 

Zahl ist die Zahl, die Sie auf- oder abrunden 

Anzahl_Stellen  gibt an, auf wie viele Dezimalstellen Sie die Zahl auf- oder abrunden möchten 
grösser 0 auf angegebene Anzahl Nachkommastellen  

 =RUNDEN( 2145.567 ; 2 )   2145.57  

gleich 0 auf ganze Zahl gerundet  

 =RUNDEN( 2145.567 ; 0 )   2146.00  

kleiner 0  auf angegebene Anzahl Vorkommastellen gerundet 
(Zehner, Hunderter, Tausender etc.) 

 =RUNDEN( 2145.567 ;-2 )   2100.00  

 

 A B C D E F 

1  Durchschnitt 

2 Nachname Test 1 Test 2 Test 3 ungerundet 1 Dezimalstelle 

3 Schüler 1 5.8 4.5 4.5 =MITTELWERT(B3:D3) =RUNDEN(E3;1) 

4 Schüler 2  4.5 5 4.75 4.8 

5 Schüler 3 5.6 5.5 5 5.366666667 5.4 

6 Schüler 4 5.6  4.5 5.05 5.1 

      oder direkt: 

7 Schüler 5 6 4.5 5  =RUNDEN(MITTELWERT(B3:D3);1) 

8 Schüler 6 4 4 3.5  3.8 

9 Schüler 7 5.2 4 5  4.7 

Tabelle 1: In Spalte F wird der Durchschnittswert aus Spalte E auf 1 Dezimalstelle gerundet; ab Zeile 7 mit einer verschachtelten 
Funktion. 

 

Beispiele für Anzahl Stellen   Zehner Hunderter Tausender 
 –1 –2 –3 
 Zehntel Hunderstel Tausendstel 
 1 2 3 

Auf Halbe runden 
Der Trick liegt darin, dass Sie zuerst den doppelten Betrag auf ganze Zahlen runden und anschliessend 
durch Zwei teilen. Beispiele: 

Betrag 4.7 4.3 5.24 5.26 

Verdoppeln 9.4 8.6 10.48 10.52 

Auf ganze Zahl runden 9.0 9.0 10.00 11.00 

Halbieren 4.5 4.5 5.00 5.50 

=RUNDEN(Zahl*2 ;0) /2 
 A B C D E 

1     Durchschnitt 

2 Nachname Test 1 Test 2 Test 3 direkt 

3 Schüler 1 5.8 4.5 4.5 =RUNDEN( MITTELWERT( B3:D3 ) * 2 ; 0 ) / 2 

4 Schüler 2  4.5 5 5 

5 Schüler 3 5.6 5.5 5 5.5 

6 Schüler 4 5.6  4.5 5 

Tabelle 2: Auf halbe Noten gerundet (direkt). 

Auf 5er runden 
Dazu verwenden Sie denselben Trick wie bei der Rundung auf Halbe. Ein 5er ist der 20ste Teil eines Gan-
zen. Deshalb multiplizieren Sie den Ausgangswert zuerst mit 20, runden ihn ganzzahlig und dividieren 
das Resultat wiederum mit 20. 
 

Betrag 2.73 8.42 0.17 0.53 

mit 20 multiplizieren 54.60 168.40 3.40 10.60 

Auf ganze Zahl runden 55.00 168.00 3.00 11.00 

durch 20 teilen 2.75 8.40 0.15 0.55 

=RUNDEN(Zahl*20;0) /20 
 A B C D E 

1 Artikel Betrag Rabatt Kaufpreis gerundet auf 5 Rappen 

2 Schreibezugsets 8.15 5% = B2 * ( 100% - C2 ) =RUNDEN( D2 * 20 ; 0 ) / 20 

3 Briefwaage 23.70 10% 21.33 21.35 

Tabelle 3: Auf 5er gerundet. 

Selbstverständlich gibt es auch andere Möglichkeiten wie = R U N D E N ( Z a h l * 2 ; 1 ) / 2  oder 
 = R U N D E N ( Z a h l / 5 ; 2 ) * 5  

Allgemein gültige Form 

Mit gewünschtem Rundungsbruchteil multiplizieren, dann auf ganze Zahl runden und durch gewünschten 
Rundungsbruchteil dividieren 

Viertelrundung = R U N D E N ( Z a h l * 4 ; 0 ) / 4  

Achtelrundung  = R U N D E N ( Z a h l * 8 ; 0 ) / 8  

Drittelrundung  = R U N D E N ( Z a h l * 3 ; 0 ) / 3  

Interessant nach einem ähnlichen Prinzip runden Sie auch auf andere Zahlenreihen 
 
3er-Reihe  =RUNDEN(Zahl/3;0)*3  
5er-Reihe  =RUNDEN(Zahl/5;0)*5 

 

Weitere Rundungsfunktionen 

=AUFRUNDEN(Zahl;Anzahl_Stellen) =ABRUNDEN(Zahl;Anzahl_Stellen) 

Diese beiden Funktionen sind gleich wie die Funktion =RUNDEN(), nur dass sie spezifisch alles auf- bezie-
hungsweise abrunden. 

=GANZZAHL(Zahl) =KÜRZEN(Zahl)  

Diese Funktionen runden auf die nächste ganze Zahl ab. Die Nachkommastellen werden abgeschnitten. 
Der Unterschied liegt in der Behandlung der negativen Zahlen: 

=GANZZAHL(4.9)  4 =KÜRZEN(4.9)  4 
=GANZZAHL(–4.9)  –5 =KÜRZEN(–4.9)  –4 

=OBERGRENZE.MATHEMATIK(Zahl;Schritt;[Modus]) 
=UNTERGRENZE.MATHEMATIK(Zahl;Schritt;[Modus]) 

Zahl ist die Zahl, die Sie auf- oder abrunden möchten 
Schritt das beim Runden zu verwendende Vielfache 
[Modus] steuert für eine negative Zahl, ob sie in Richtung des grösseren oder kleineren Werts gerundet wird 

Beispiele:  =OBERGRENZE.MATHEMATIK(2.71;0.05)  2.75 
=UNTERGRENZE.MATHEMATIK(2.74;0.05)  2.70 

=VRUNDEN(Zahl;Vielfaches) nicht verwenden, da fehlerhaft! 

Zahl ist die Zahl, die Sie auf- oder abrunden möchten 
Vielfaches ist das Vielfache, auf das Sie Zahl runden möchten 

Beispiele =VRUNDEN(Zahl;6) auf Zahlen in 6er-Reihe 
=VRUNDEN(Zahl;0.5) auf Halbe 
=VRUNDEN(Zahl;0.05) auf 5 Rappen fehlerhaft; z. B. 2.675, 4.225, 6.225 etc. 


